
HP-35s Calculator Program Radiations 1

Calculate a Radiation from Two Offsets

Programmer: Dr. Bill Hazelton

Date: March, 2008.

Line Instruction Display User Instructions
O001 LBL O   LBL  O
O002 CLSTK   CLEAR  5
O003 FS?  10   FLAGS  3  .0
O004 GTO  O008
O005 SF  1   FLAGS  1  1
O006 SF  10   FLAGS  1  .0
O007 GTO  O009
O008 CF  1   FLAGS  2  1
O009 RAD  FM  OFFSETS (Key in using EQN RCL R, RCL A, etc.)
O010 PSE   PSE
O011 ENTER RIGHT AZ (Key in using EQN RCL E, RCL N, etc.)
O012 PSE   PSE
O013 INPUT  R R?   INPUT  R
O014 RCL  R
O015 HMS→   HMS→
O016 STO  A   STO  A
O017 ENTER LEFT AZ (Key in using EQN RCL E, RCL N, etc.)
O018 PSE   PSE
O019 INPUT  L L?   INPUT  L
O020 RCL  A
O021 RCL  L
O022 HMS→   HMS→
O023 –
O024 STO  E   STO  E
O025 ENTER RIGHT OS (Key in using EQN RCL E, RCL N, etc.)
O026 PSE   PSE
O027 INPUT  R R?   INPUT  R
O028 ENTER LEFT OS (Key in using EQN RCL E, RCL N, etc.)
O029 PSE   PSE
O030 INPUT  L L?   INPUT  L
O031 RCL  R
O032 RCL  E
O033 SIN
O034 ×
O035 RCL  E
O036 COS
O037 RCL×  R
O038 RCL+  L
O039 ÷

Line Instruction Display User Instructions
O040 ATAN   ATAN
O041 STO  C   STO  C
O042 RCL  A
O043 x < > y
O044 –
O045 →HMS   →HMS
O046 STO  B   STO  B
O047 RADIATION AZ (Key in using EQN RCL R, RCL A, etc.)
O048 PSE   PSE
O049 VIEW  B   VIEW  B
O050 RCL  R
O051 RCL  C
O052 SIN
O053 ÷
O054 STO  D   STO  D
O055 RAD LENGTH (Key in using EQN RCL R, RCL A, etc.)
O056 PSE   PSE
O057 VIEW  D   VIEW  D
O058 PROGRAM  END (Key in using EQN RCL P, RCL R, etc.)
O059 PSE   PSE
O060 FS?  1   FLAGS  3  1
O060 CF  10   FLAGS  2  .0
O060 RTN   RTN

Notes

(1) This program allows a radiation (azimuth and distance) to an object to be
computed from two offsets measured from two lines of known azimuth to the
object.

(2) The two lines from which the offsets are measured do not have to be orthogonal.
The closer they are to orthogonal, the better, but the program will work with any
realistic set of measurements. It will also work across the 0° line, and with
traverse lines in different quadrants.

(3) The purpose of the program is to facilitate conversion of older survey data, in
which corners were often located by pairs of offsets, to a form in which a
radiation is employed. This will simplify calculations based on traverses and
radiations, i.e., vectors.

(4) Azimuths are entered and displayed in HP notation, i.e., DDD.MMSS, at all
times.

(5) Feet or meters (or any other linear units) can be used, provided their use is
consistent.

(6) The original code for this program was developed by Philip R. Price at the State
Rivers and Water Supply Commission (SR&WSC) Survey Branch, Victoria,
Australia, in November, 1976, for the HP-25 calculator. This program is an
update and adaptation for the HP-35s, but is based on Phil Price’s original
solution and HP-25 implementation.

Theory

If an object, commonly a corner, is located by offsets, these can be converted to the equivalent
radiation using the program. In the example below, the corner was located by two offsets, being
5.523 on the right and 25.408 on the left. The offsets are measured at right angles to their
respective traverse lines, and are the distance between the object and the traverse line. The right
traverse line has an azimuth of 91° 56', while the left traverse line has an azimuth of 340° 36'. See
the figure below:

Generalizing the diagram and giving it symbols, the general situation becomes as follows:

The angle θ is the difference between the two known azimuths, Azleft and Azright. The offset from
the right traverse line to the object is dright, while the offset from the left traverse line to the object
is dleft. Because the offsets are measured at right angles to the traverse lines, their azimuths are
also known.

The angle α is the angle between the right traverse line and the radiation, and is determined using
the following formula:

α = arctan
dright sin θ

dleft + dright cos θ
⎛ 

⎝ ⎜ 
⎞ 

⎠ ⎟ 

Subtracting α from Azright gives the azimuth of the radiation, Azrad. The length of the radiation,
drad, is derived using:

� 

drad =
dright
sinα

Precision of Results
Note that the precision of the derived azimuth and distance is limited by the precision with which
the offsets were measured, as well as the geometry of the offsets and radiation. The program
provides the one solution given the data, but without redundant measurements the precision of the
solution is unknown. It would be unwise to use azimuths for the radiation that are more precise
that about one minute of arc. Experiment with changing the values of the offsets by small
amounts that are consistent with their probable precision, as well as changing the azimuths by
amounts consistent with their precision over the distances to the offsets, and see what happens to
the azimuth and distance values for the radiation.

Sample Computations

1. Using the example given, where the right azimuth is 91° 56', the left azimuth is 340° 36',
the right offset is 5.523 and the left offset is 25.408, the radiation has an azimuth of 79°
32' and a distance of 25.72.

2. Using a right azimuth of 160° 36' and a left azimuth of 91° 56', a right offset of 13.272
and a left offset of 15.693, the radiation has an azimuth of 129° 32' and a distance of
25.72.

3. Using a right azimuth of 268° 35', a left azimuth of 195° 12', a right offset of 11.782 and
a left offset of 9.467, the radiation has an azimuth of 227° 15' and a distance of 17.84.

4. Using a right azimuth of 340° 36', a left azimuth of 271° 56', a right offset of 11.57 and a
left offset of 8.78, the radiation has an azimuth of 300° 55' and a distance of 18.12.

Running the Program
With everything to hand, press XEQ  O, then press ENTER.

The calculator briefly displays  RAD  FM  OFFSETS, then  briefly displays ENTER RIGHT AZ,
then prompts R?

Key in the value of the azimuth of the right traverse line, in HP notation (DDD.MMSS).
Press R/S.

The calculator briefly displays ENTER LEFT AZ, then prompts L?

Key in the value of the azimuth of the left traverse line, in HP notation (DDD.MMSS).
Press R/S.

The calculator briefly displays ENTER RIGHT OS, then prompts R?

Key in the value of the right offset. Press R/S.

The calculator briefly displays ENTER LEFT OS, then prompts L?

Key in the value of the left offset. Press R/S.

The calculator briefly displays RADIATION AZ, then shows B= and the azimuth in HP notation
(DDD.MMSS). Press R/S.

The calculator briefly displays RAD LENGTH, then shows D= and the length of the radiation.

Press R/S, the calculator briefly displays PROGRAM END, completes the program and resets
Flag 10 to its state at the start of the program. The azimuth (in HP notation) and the distance of
the radiation remain on the stack, in the Y and X locations, respectively.

To compute additional radiations from offset, press XEQ  O, then press ENTER, and start the
program again.

Storage Registers Used

A Azimuth of right traverse line (in decimal degrees)
B Azimuth of the radiation (in HP notation, DDD.MMSS)
C α, the angle between the right traverse line and the radiation (in decimal degrees)
D Length of the radiation
E θ, the angle between the traverse lines (in decimal degrees)
L Left traverse azimuth (in HP notation) or left offset (temporary storage)
R Right traverse azimuth (in HP notation) or right offset (temporary storage)

Labels Used

Label O Length  =   287 Checksum  =  55B5

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computations to check proper operation after entry.

Flags Used

Flags 1 and 10 are used by this program. Flag 10 is set for this program, so that equations can be
shown as prompts. Flag 1 is used to record the setting of Flag 10 before the program begins. At
the end of the program, Flag 10 is reset to its original value, based on the value in Flag 1.


